TEMPLATE = lib
TARGET = sea
CONFIG += staticlib
CONFIG += console
VERSION = 2.0.1
OBJECTS_DIR=$$PWD/core/_obj


win32 {
    DESTDIR = $$PWD/lib
}

!symbian: unix {
    DESTDIR = $$PWD/lib
}


INCLUDEPATH = $$PWD/core/inc $$PWD/core/src
DEPENDPATH = $$PWD/core/inc $$PWD/core/src

SOURCES = \
    $$PWD/core/src/sea_object.cpp \
    $$PWD/core/src/sea_link.cpp \
    $$PWD/core/src/sea_list.cpp \
    $$PWD/core/src/sea_association.cpp \
    $$PWD/core/src/sea_event.cpp \
    $$PWD/core/src/sea_platform.cpp \
    $$PWD/core/src/sea_atomic_op.cpp \
    \
    $$PWD/core/src/sea_queue.cpp \
    $$PWD/core/src/sea_tree.cpp \
    $$PWD/core/src/sea_stack.cpp \
    \
    $$PWD/core/src/sea_engine.cpp \
    $$PWD/core/src/sea_manager.cpp \
    $$PWD/core/src/sea.cpp \
    $$PWD/core/src/sea_parser.cpp \
    $$PWD/core/src/sea_meta_system.cpp \
    \
    $$PWD/core/src/sea_timer.cpp \
    $$PWD/core/src/sea_timer_server.cpp \
    \
    $$PWD/core/src/sea_searchengine.cpp \
    \
    $$PWD/core/src/sea_memtracker.cpp

HEADERS = \
    ./core/inc/sea_consts.h \
    ./core/inc/sea_config.h \
    ./core/inc/sea_assert.h \
    ./core/inc/sea_object.h \
    ./core/inc/sea_link.h \
    ./core/inc/sea_list.h \
    ./core/inc/sea_association.h \
    ./core/inc/sea_event.h \
    ./core/inc/sea_iterator.h \
    ./core/inc/sea_platform.h \
    ./core/inc/sea_atomic_op.h \
    \
    ./core/inc/sea_types.h \
    ./core/inc/sea_defines.h \
    \
    ./core/inc/sea_queue.h \
    ./core/inc/sea_tree.h \
    ./core/inc/sea_stack.h \
    \
    ./core/inc/sea.h \
    ./core/inc/sea_engine.h \
    ./core/inc/sea_manager.h \
    ./core/inc/sea_parser.h \
    ./core/inc/sea_meta_system.h \
    \
    ./core/inc/sea_timer.h \
    ./core/inc/sea_timer_server.h \
    \
    ./core/inc/sea_searchengine.h \
    \
    ./core/inc/sea_memtracker.h


contains( CONFIG, qt ) {
    message( Platform: SEA_OS_QT )

    SOURCES += \
        $$PWD/core/src/sea_platform_qt.cpp
    HEADERS += \
        ./core/inc/sea_platform_qt.h
} else {

    win32 {
        message( Platform: SEA_OS_WIN32 )

        SOURCES += \
            $$PWD/core/src/sea_platform_win.cpp
        HEADERS += \
            ./core/inc/sea_platform_win.h
    }

    symbian {
        message( Platform: SEA_OS_SOS )
    
        SOURCES += \
            $$PWD/core/src/sea_platform_sos.cpp
        HEADERS += \
            ./core/inc/sea_platform_sos.h
    }
    else {
        unix {
            message( Platform: SEA_OS_LINUX )
        
            SOURCES += \
                $$PWD/core/src/sea_platform_posix.cpp
            HEADERS += \
                ./core/inc/sea_platform_posix.h
        }
    }
}

include( sea.pri )


